//
//    This is a project funding experiment:
//
//    https://support.buidlguidl.com
//
//    Support the BuidlGuidl! Contact: @austingriffith
//
//

// Sources flattened with buidler v1.4.8 https://buidler.dev

// File @openzeppelin/contracts/GSN/Context.sol@v3.2.0

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with GSN meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address payable) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes memory) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}


// File @openzeppelin/contracts/access/Ownable.sol@v3.2.0

// SPDX-License-Identifier: MIT

pragma solidity ^0.6.0;

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor () internal {
        address msgSender = _msgSender();
        _owner = msgSender;
        emit OwnershipTransferred(address(0), msgSender);
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        require(_owner == _msgSender(), "Ownable: caller is not the owner");
        _;
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions anymore. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby removing any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        emit OwnershipTransferred(_owner, address(0));
        _owner = address(0);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        emit OwnershipTransferred(_owner, newOwner);
        _owner = newOwner;
    }
}


// File contracts/MVPCLR.sol

pragma solidity >=0.6.6 <0.7.0;

contract MVPCLR is Ownable {

  uint256 public roundStart;
  uint256 public immutable roundDuration;
  uint256 public recipientCount = 0;

  event RoundStarted(uint256 roundStart, uint256 roundDuration);
  event RecipientAdded(address addr, bytes32 data, string link, uint256 index);
  event Donate(address sender, uint256 value, uint256 index);
  event MatchingPoolDonation(address sender, uint256 value);
  event Distribute(address to, uint256 index, uint256 amount);

  modifier beforeRoundOpen() {
    require(roundStart == 0, "MVPCLR:beforeRoundOpen - Round already opened");
    _;
  }

  modifier isRoundOpen() {
    require(
      getBlockTimestamp() < (roundStart + roundDuration),
      "MVPCLR:isRoundOpen - Round is not open"
    );
    _;
  }

  modifier isRoundClosed() {
    require(
      roundStart != 0 &&
      getBlockTimestamp() >= (roundStart + roundDuration),
      "MVPCLR:isRoundClosed Round is not closed"
    );
    _;
  }

  constructor(uint256 _roundDuration) public {
    require(_roundDuration < 31536000,"MVPCLR: round duration too long");
    roundDuration = _roundDuration;
  }

  function startRound()
  public
  onlyOwner
  beforeRoundOpen
  {
    roundStart = getBlockTimestamp();
    emit RoundStarted(roundStart, roundDuration);
  }

  function getBlockTimestamp() public view returns (uint256) {
    return block.timestamp;
  }

  function addRecipient(address payable addr, bytes32 data, string memory link)
  public
  onlyOwner
  beforeRoundOpen
  {
    emit RecipientAdded(addr, data, link, recipientCount++);
  }

  function donate(uint256 index) public payable isRoundOpen {
    require(index<recipientCount, "CLR:donate - Not a valid recipient");
    emit Donate(_msgSender(), msg.value, index);
  }

  function distribute(address payable to, uint256 index, uint256 amount)
  external
  onlyOwner
  isRoundClosed
  {
    to.transfer(amount);
    emit Distribute(to,index,amount);
  }


  // receive donation for the matching pool
  receive() external payable {
    require(roundStart == 0 || getBlockTimestamp() < roundStart + roundDuration, "CLR:receive closed");
    emit MatchingPoolDonation(_msgSender(), msg.value);
  }

}